Appendix A4. Simulated annealing approach.
In practice, simulated annealing involves making small, random perturbations to the current state and comparing the new state with the old state in terms of distance from a pre-specified calibration target. The simulated annealing algorithm was developed in Stata, and random perturbations were made by selecting one transition (e.g., from abstinent to low risk) at random and increasing or decreasing it by one percentage point in all 24 transition matrices representing different demographic groups. The remaining elements in the transition’s row (i.e., the other possible transitions from the same starting state) were also increased or decreased proportionally to maintain the stochasticity of the matrix (i.e., rows sum to 1 with no negative elements). 

The characteristic that distinguishes simulated annealing from simpler optimization methods (e.g., hill climber algorithms) is that the algorithm can accept a perturbation that increases distance to the target with a probability that is a function of distance from the target and system “temperature.” In our algorithm, the temperature parameter was set to 10 at the beginning of the simulated annealing process and declined exponentially (i.e., the temperature was multiplied by 0.995 each iteration) (Kirkpatrick et al., 1983). If the predicted distribution resulted in a larger distance to the actual distribution than the previous iteration, then the change was accepted and carried forward to the next iteration with probability p:  

where D0 is the distance to the target before the perturbation, D1 is the distance to the target after the perturbation, and T is the current temperature of the system. This implies that perturbations that produce greater distance to the target than its predecessor matrices are more likely to be accepted during earlier iterations when the temperature is high. This feature of simulated annealing is designed to allow the algorithm to approach a global minimum distance rather than getting “stuck” in a local minimum (Kong et al., 2009). The simulated annealing algorithm had a stopping rule when temperature fell below 0.01 (Vanni et al., 2011). 

Simulated Annealing Code (Stata):
*Simulated Annealing - modify all matrices in the same way

*Note: This is the code for the first calibration (to generate 3- to 8-year matrices). We use the same approach for 8–16 years.

/*Key parameters: 

   *Distance measure: Squared differences

   *Calibration target: Midpoint between observed and predicted (using short-term TPs) state distribution

   *Initial temperature: 10

   *Temperature delta: Multiply by .995

   *unit of change: 0.01 - 1 percent

*/

   *Calculate initial distance to target

   local distance = 0

   forvalues y = 1/5 {

      local tmp = 100*(targets[1,`y'] - current[1,`y'])

      local tmp2 = `tmp'*`tmp'

      local distance = `distance' + `tmp2'

   }

   local best = `distance'

   local temperature = 10

   mat best = current

   *Create copies of all of the coefficient matrices

   local matrices y1age3_fem1_raeth4_stata y1age3_fem1_raeth3_stata y1age3_fem1_raeth2_stata y1age3_fem1_raeth1_stata y1age3_fem0_raeth4_stata y1age3_fem0_raeth3_stata y1age3_fem0_raeth2_stata y1age3_fem0_raeth1_stata y1age2_fem1_raeth4_stata y1age2_fem1_raeth3_stata y1age2_fem1_raeth2_stata y1age2_fem1_raeth1_stata y1age2_fem0_raeth4_stata y1age2_fem0_raeth3_stata y1age2_fem0_raeth2_stata y1age2_fem0_raeth1_stata y1age1_fem1_raeth4_stata y1age1_fem1_raeth3_stata y1age1_fem1_raeth2_stata y1age1_fem1_raeth1_stata y1age1_fem0_raeth4_stata y1age1_fem0_raeth3_stata y1age1_fem0_raeth2_stata y1age1_fem0_raeth1_stata

   foreach x of local matrices {

      mat `x'_copy = `x'

   }

   mat input sa = (., `distance' `distance')

   *Start Algorithm

   while `temperature' > 0.01 {   
      *reload fresh data at each iteration (no changes that we wish to preserve are made to the dataset, only the TP matrices)

      use ../modified/simy3y8.dta, clear

      *Choose a random element (row and column) to tweak in a random direction (add or subtract)

      local randRow = floor(5*runiform()+1)

      local randElement = floor(5*runiform()+1)

      local add = round(runiform())

      *Save the matrix in case we reject

      foreach x of local matrices {

         mat `x'_bkmk = `x'_copy

      }

      if `add' == 0 { /*subtract*/

         foreach x of local matrices {

            local origval = `x'_copy[`randRow',`randElement']

            local change = min(`x'_copy[`randRow',`randElement'], .01)

            mat `x'_copy[`randRow',`randElement'] = `x'_copy[`randRow',`randElement']-`change'

            forvalues y = 1/5 {

               if `y' != `randElement' {

                  mat `x'_copy[`randRow',`y'] = `x'_copy[`randRow',`y'] + `change'*`x'_copy[`randRow',`y']/(1-`origval')

               }

            }

         }

      }

      else if `add' == 1 { /*add*/

         foreach x of local matrices {

            local origval = `x'_copy[`randRow',`randElement']

            local change = min(1-`x'_copy[`randRow',`randElement'], .01)

            mat `x'_copy[`randRow',`randElement'] = `x'_copy[`randRow',`randElement']+`change'

            forvalues y = 1/5 {

               if `y' != `randElement' {

                  mat `x'_copy[`randRow',`y'] = `x'_copy[`randRow',`y'] - `change'*`x'_copy[`randRow',`y']/(1-`origval')

               }

            }

         }

      }

      *Choose the right matrix (based on age, gender, race/eth) for each person and cycle 5 times (year 3 to year 8)

      *Note: age_4 is the CS participant's age at the 3y interview. 

         sort id

         assert id==_n

         forvalues y = 1/`m' {

            assert `y'==id[`y']

            local f = female[`y']

            local r = raceeth4[`y']

            local id = id[`y']

            if ds4[`y']!=. & ds5[`y']!=. & weight38!=. & weight38!=0 {

               mat f3`id' = s3`id'

               forvalues z = 0/4 {

                  qui: gen agetmp = age_4+`z' if id==`y'

                  qui: recode agetmp (18/35 = 1) (36/50 = 2) (51/999 = 3) (nonmissing = . ) (missing = .), gen(agetmp3)

                  local a = agetmp3[`y']

                  mat f3`id'=f3`id'*y1age`a'_fem`f'_raeth`r'_stata_copy

                  mat rownames f3`id' = `id'

                  drop agetmp

                  drop agetmp3

               }

            }

         }

      *At the end, each person has a 1x5 vector representing probabilities of being in each state. Put them together and save to dataset.

         sort id

         assert id==_n

         mat input fin = ()

         forvalues y = 1/`m' {

            local id = id[`y']

            if ds4[`y']!=. & ds5[`y']!=. & weight38[`y']!=. & weight38[`y']!=0 {

               mat tmp = f3`y', `id'

               mat fin = fin \ tmp

            }

            mat colnames fin = predAbs predLow predMed predHigh predVigh id

         }

         tempfile t1

         qui: save `t1'

         clear

         qui: svmat fin, names(col)

         sort id

         tempfile t2

         qui: save `t2'

      *Merge the matrix of individual's vectors onto the dataset with demographic characteristics, look at actual vs. predicted

         use `t1'

         keep id age_4 female raceeth4 ds* weight38   
         qui: keep if weight38!=. & weight38!=0

         sort id

         qui: merge 1:1 id using `t2'

         assert _merge ==3

         drop _merge

         qui: tab ds5, gen(ds5_)

         rename ds5_1 actAbs

         rename ds5_2 actLow

         rename ds5_3 actMed

         rename ds5_4 actHigh

         rename ds5_5 actVHigh

         qui: mean act* [pw=weight38]

         mat actual = e(b)

         qui: mean pred* [pw=weight38]

         mat pred = e(b)

         mat current = pred

         local newdistance = 0

         forvalues y = 1/5 {

            local tmp = 100*(targets[1,`y'] - current[1,`y'])

            local tmp2 = `tmp'*`tmp'

            assert (`tmp2') > 0

            local newdistance = `newdistance' + `tmp2'

         }

         *Always accept if smaller distance

         if `newdistance' < `distance' {

            local distance = `newdistance'

            if `distance' < `best' {

               local best = `distance'

               foreach x of local matrices {

                  mat `x'_best = `x'_copy

               }

               mat best = current

            }

         } 

         *Accept with probability as a function of distance (how much worse) and temperature

         else {

            local prob = exp((`distance' - `newdistance')/`temperature')

            local rand = runiform()

            if `prob' > `rand' {

               local distance = `newdistance'

            }

            else {

               foreach x of local matrices {

                  mat `x'_copy = `x'_bkmk

               }

            }

         }

   di "Temp: `temperature'; Current Distance: `distance'; Best Distance `best'"

   mat sa = sa\(`temperature', `distance', `best')

   local temperature = `temperature' * .995
